
Corrections required for test case 7.1.4.12

Change 1

	Testcase name
	7.1.4.12

	Reason for change
	During the handover, handling of HARQ Errors must be OFF in SS. Since this is intrafrequency handover the handover message on Cell 1 will not be RLC acked as the UE would have moved to Cell 2. This can cause a problem when HARQ Error Reporting is switched on immediately after handover as the SS will keep on sending the Handover RLC PDU for some time. In addition the processing of some SYSTEM_REQ messages from handover finction(inside the function f_EUTRA_508RRC_IntraLTE_HO_InterCell)  can take time as are not confirmed and are activated now, but the TTCN core already enters the next fuction f_EUTRA_HarqErrorHandling_Set which enables HARQ Errors.

Therefore an optional parameter "boolean

p_CnfFlag := tsc_NoCnfReq"  can be handed over to

f_EUTRA_508RRC_IntraLTE_HO_InterCell replacing the already existing

variable v_CnfFlag; in case of f_TC_7_1_4_12_EUTRA this parameter can be

set to true.

	Summary of change
	In function f_EUTRA_508RRC_IntraLTE_HO_InterCell replace variable v_CnfFlag with p_CnfFlag := tsc_NoCnfReq. For f_TC_7_1_4_12_EUTRA set the variable v_CnfFlag as True.

	Source of change
	\7_1\MAC_714.ttcn


Before change:

	function f_TC_7_1_4_12_EUTRA() runs on EUTRA_PTC

…

    //@siclog "Step 7" siclog@

    //@sic R5s100240 R5s100772 sic@

    f_EUTRA_HarqErrorHandling_Set(ignore); // Stop 'Harq Errors not tolerant, UE may not send an HARQ ack on source cell, and SS may deect a DTX as NACK.

    f_EUTRA_508RRC_IntraLTE_HO_InterCell (eutra_Cell1, eutra_Cell2, v_DrbConfigList, -, -, -, cs_UL_GrantConfig_Stop);
    // to enable reception of SR in cell 2

    //@sic R5s110456 R5s120023 sic@

    f_EUTRA_HarqErrorHandling_Set(raiseInconc); // Restart 'Harq Errors not tolerant'

    f_SS_ConfigSchedulingRequestIndMode(eutra_Cell2, tsc_L1Mac_IndicationMode_Enable);

    //@siclog "Step 8" siclog@

    t_Watchdog.start(2.0);

    alt {

      []SYSIND.receive(car_SchedulingReq_IND (eutra_Cell2, ?))

        {

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8: UE is  sending SR ");

          t_Watchdog.stop;

        }

      [] t_Watchdog.timeout

        {

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");

        }

    }

    f_SS_ConfigPhich_TestMode(eutra_Cell2, cs_Phich_ExplicitMode(cs_HARQ_ModeList_28Nack));

    //@siclog "Step 9" siclog@

    DRB.send (cas_DRB_COMMON_REQ_PDCP_SDU (eutra_Cell2,tsc_RbId_DRB2, cs_TimingInfo_Now,  cs_PDCP_SDUList_1 (crs_PDCP_SDU_L39_1)));

    //@siclog "Step 10" siclog@

    t_Watchdog.start;

    SYSIND.receive(car_SchedulingReq_IND (eutra_Cell2, ?));

    t_Watchdog.stop;

    //@siclog "Step 11" siclog@

    // Automatically taken care off

    //@siclog "Step 12" siclog@

    t_Watchdog.start;

    DRB.receive (car_DRB_COMMON_IND_PDCP_SDUList (eutra_Cell2,

                                                  tsc_RbId_DRB2,

                                                  cr_TimingInfo_Any,

                                                  cr_PDCP_SDUList_1(crs_PDCP_SDU_L39_1)));

    t_Watchdog.stop;

    f_SS_ConfigSchedulingRequestIndMode (eutra_Cell2, tsc_L1Mac_IndicationMode_Disable);

    // stop further SS reception reporting

    // Note even though UE may potentially do multiple retransmissions, the corresponding UL PDCP SDU's are not sent to

    // TTCN due to duplicate detection in SS RLC/PDCP.

    //@siclog "Step 13" siclog@

    // Automatically taken care off

    //@siclog "Step 13A" siclog@

    v_CellPowerList := {

      cs_CellPower (eutra_Cell1, -85),          //@sic R5-106557 sic@

      cs_CellPower (eutra_Cell2, -91)           //@sic R5-106557 sic@

    };

    f_EUTRA_SetCellPowerList(v_CellPowerList);  // @sic R5s110726 additional changes for change 2 sic@

    //@siclog "Step 14-15" siclog@

    //@sic R5s100240 R5s120023 sic@

    f_EUTRA_HarqErrorHandling_Set(ignore); // Stop 'Harq Errors not tolerant, UE may not send an HARQ ack on source cell, and SS detects a DTX as NACK.

    f_EUTRA_508RRC_IntraLTE_HO_InterCell (eutra_Cell2,

                                          eutra_Cell1,

                                          v_DrbConfigList,

                                          tsc_C_RNTI_Def);
    f_EUTRA_HarqErrorHandling_Set(raiseInconc); // Restart 'Harq Errors not tolerant'

    //@siclog "Step 16" siclog@




After change:

	  function f_TC_7_1_4_12_EUTRA() runs on EUTRA_PTC

….

    //@siclog "Step 7" siclog@

    //@sic R5s100240 R5s100772 sic@

    f_EUTRA_HarqErrorHandling_Set(ignore); // Stop 'Harq Errors not tolerant, UE may not send an HARQ ack on source cell, and SS may deect a DTX as NACK.

    f_EUTRA_508RRC_IntraLTE_HO_InterCell (eutra_Cell1, eutra_Cell2, v_DrbConfigList, -, -, -, cs_UL_GrantConfig_Stop, -, -, -, true);
    // to enable reception of SR in cell 2

    //@sic R5s110456 R5s120023 sic@

    f_EUTRA_HarqErrorHandling_Set(raiseInconc); // Restart 'Harq Errors not tolerant'

    f_SS_ConfigSchedulingRequestIndMode(eutra_Cell2, tsc_L1Mac_IndicationMode_Enable);

    //@siclog "Step 8" siclog@

    t_Watchdog.start(2.0);

    alt {

      []SYSIND.receive(car_SchedulingReq_IND (eutra_Cell2, ?))

        {

          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8: UE is  sending SR ");

          t_Watchdog.stop;

        }

      [] t_Watchdog.timeout

        {

          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 8");

        }

    }

    f_SS_ConfigPhich_TestMode(eutra_Cell2, cs_Phich_ExplicitMode(cs_HARQ_ModeList_28Nack));

    //@siclog "Step 9" siclog@

    DRB.send (cas_DRB_COMMON_REQ_PDCP_SDU (eutra_Cell2,tsc_RbId_DRB2, cs_TimingInfo_Now,  cs_PDCP_SDUList_1 (crs_PDCP_SDU_L39_1)));

    //@siclog "Step 10" siclog@

    t_Watchdog.start;

    SYSIND.receive(car_SchedulingReq_IND (eutra_Cell2, ?));

    t_Watchdog.stop;

    //@siclog "Step 11" siclog@

    // Automatically taken care off

    //@siclog "Step 12" siclog@

    t_Watchdog.start;

    DRB.receive (car_DRB_COMMON_IND_PDCP_SDUList (eutra_Cell2,

                                                  tsc_RbId_DRB2,

                                                  cr_TimingInfo_Any,

                                                  cr_PDCP_SDUList_1(crs_PDCP_SDU_L39_1)));

    t_Watchdog.stop;

    f_SS_ConfigSchedulingRequestIndMode (eutra_Cell2, tsc_L1Mac_IndicationMode_Disable);

    // stop further SS reception reporting

    // Note even though UE may potentially do multiple retransmissions, the corresponding UL PDCP SDU's are not sent to

    // TTCN due to duplicate detection in SS RLC/PDCP.

    //@siclog "Step 13" siclog@

    // Automatically taken care off

    //@siclog "Step 13A" siclog@

    v_CellPowerList := {

      cs_CellPower (eutra_Cell1, -85),          //@sic R5-106557 sic@

      cs_CellPower (eutra_Cell2, -91)           //@sic R5-106557 sic@

    };

    f_EUTRA_SetCellPowerList(v_CellPowerList);  // @sic R5s110726 additional changes for change 2 sic@

    //@siclog "Step 14-15" siclog@

    //@sic R5s100240 R5s120023 sic@

    f_EUTRA_HarqErrorHandling_Set(ignore); // Stop 'Harq Errors not tolerant, UE may not send an HARQ ack on source cell, and SS detects a DTX as NACK.

    f_EUTRA_508RRC_IntraLTE_HO_InterCell (eutra_Cell2,

                                          eutra_Cell1,

                                          v_DrbConfigList,

                                          tsc_C_RNTI_Def, -, -, -, -, -, -, true);
    f_EUTRA_HarqErrorHandling_Set(raiseInconc); // Restart 'Harq Errors not tolerant'

    //@siclog "Step 16" siclog@

…


Change 2

	Testcase name
	7.1.4.12

	Reason for change
	During the handover, handling of HARQ Errors must be OFF in SS. Since this is intrafrequency handover the handover message on Cell 1 will not be RLC acked as the UE would have moved to Cell 2. This can cause a problem when HARQ Error Reporting is switched on immediately after handover as the SS will keep on sending the Handover RLC PDU for some time. In addition the processing of some SYSTEM_REQ messages from handover finction(inside the function f_EUTRA_508RRC_IntraLTE_HO_InterCell)  can take time as are not confirmed and are activated now, but the TTCN core already enters the next fuction f_EUTRA_HarqErrorHandling_Set which enables HARQ Errors.

Therefore an optional parameter "boolean

p_CnfFlag := tsc_NoCnfReq"  can be handed over to

f_EUTRA_508RRC_IntraLTE_HO_InterCell replacing the already existing

variable v_CnfFlag; in case of f_TC_7_1_4_12_EUTRA this parameter can be

set to true.

	Summary of change
	In function f_EUTRA_508RRC_IntraLTE_HO_InterCell replace variable v_CnfFlag with p_CnfFlag := tsc_NoCnfReq. For f_TC_7_1_4_12_EUTRA set the variable v_CnfFlag as True.

	Source of change
	CommonEUTRA\EUTRA_RRCSteps.ttcn


Before change:

	  function f_EUTRA_508RRC_IntraLTE_HO_InterCell ( CellId_Type                           p_SourceCellId,

                                                  CellId_Type                           p_TargetCellId,

                                                  template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def,

                                                  C_RNTI                                p_C_RNTI        := tsc_C_RNTI_Def2,

                                                  template (omit)  MeasConfig           p_MeasConfig    := omit,

                                                  template (omit)  DciUlInfo_Type       p_DciUlInfo_TargetCell      := cs_DciInfo_CcchDcchDtchUL_Def,

                                                  template (value) UL_GrantConfig_Type  p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_Def,

                                                  template (value) UL_GrantConfig_Type  p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_Def,

                                                  Dsr_TransMax_Type                     p_Dsr_TransMax := n4,

                                                  template (omit) CarrierFreqEUTRA      p_CarrierFreqEutra := omit )  runs on EUTRA_PTC

  { /* @sic R5s100491: enhancement to support MeasConfig sic@ */

    /* @sic R5s100524: common change of HO functions sic@ */

    /* @sic R5-106063: Sequence of inter-cell handover sic@ */

    /* @sic R5s100801: No CNF for SS configuration of steps 10..14 sic@

     *                 when there is data pending at the UE the UE may send this data immediately after the HO;

     *                 this would collide with CNFs sent during reconfigurations of source and target call after the HO;

     *                 on the other hand none of the reconfigurations in source or target cell should have impact in cases the test case sends data after the HO

     *                 => reconfiguration steps 10..14 are done without request of CNF as proposed in the CR */

    /* @sic R5s100772: p_UL_GrantConfig_SourceCell sic@ */

    /* @sic R5s110168: additional parameter p_Dsr_TransMax sic@ */

    /* @sic R5-112122: UL_DL_Earfcn is omit in case of inter-cell HO when there is no change of frequency sic@ */

    /* @sic R5s110279 change 1: new optional parameter p_CarrierFreqEutra to force the CarrierFreqEutra e.g. to omit the ul-CarrierFreq as requested for TC_8_2_4_9 sic@ */

    /* @sic R5s110743, R5s120039: handling of MeasGapConfig sic@ */

    /* @sic R5s120057, R5-120677: In case of multiband scenarios the bandwidth can be different for source and target cell sic@ */

    /* @sic R5-120721 change 7: handover of PDCP Count for AM DRBs from source cell to target cell sic@ */

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment

    var Frequency_IE_Type v_Frequency_IE_SourceCell := f_EUTRA_CellInfo_GetFrequencyIEs (p_SourceCellId);

    var Frequency_IE_Type v_Frequency_IE_TargetCell := f_EUTRA_CellInfo_GetFrequencyIEs (p_TargetCellId);

    var CarrierFreqEUTRA v_CarrierFreqEutra_SourceCell := v_Frequency_IE_SourceCell.UL_DL_Earfcn;

    var CarrierFreqEUTRA v_CarrierFreqEutra_TargetCell := v_Frequency_IE_TargetCell.UL_DL_Earfcn;

    var template (omit) CarrierFreqEUTRA v_CarrierFreqEutra := omit;

    var template (omit) CarrierBandwidthEUTRA v_CarrierBandwidthEUTRA := omit;

    var template (value) MobilityControlInfo v_MobilityControlInfo;

    var template (value) RadioResourceConfigDedicated v_RadioResourceConfigDedicated := f_Generate_cs_508_RadioResourceConfigDedicated_HO (p_TargetCellId, p_Dsr_TransMax);

    var boolean v_CnfFlag := tsc_NoCnfReq;  // @sic R5s100801: could be changed to an optional parameter if necessary sic@

    var boolean v_MeasGapConfigIsPresent := isvalue(p_MeasConfig) and ispresent(p_MeasConfig.measGapConfig);

    var boolean v_PucchConfigCommon_Needed := false;

    var DrxCtrl_Type v_DrxCtrl;

    if (isvalue(p_CarrierFreqEutra)) {      // use the give CarrierFreqEutra; @sic R5s110279 change 1 sic@

      v_CarrierFreqEutra := p_CarrierFreqEutra;

    }

    else {                                  // => determine the CarrierFreqEutra depending on cell information stored for sourece and target cell

      if (v_CarrierFreqEutra_SourceCell != v_CarrierFreqEutra_TargetCell) {    // only if CarrierFreqEUTRA is different in both cells the CarrierFreqEUTRA of the target cell is handed over @sic R5-112122 sic@

        v_CarrierFreqEutra := v_CarrierFreqEutra_TargetCell;

      }

    }

    if (v_Frequency_IE_SourceCell.DL_ChBandwidth != v_Frequency_IE_TargetCell.DL_ChBandwidth) {                         // @sic R5s120057 sic@

      v_PucchConfigCommon_Needed := true;                                                                               // @sic R5s120209 change 1 sic@

      v_CarrierBandwidthEUTRA := cs_CarrierBandwidthEUTRA(v_Frequency_IE_TargetCell.DL_ChBandwidth);

    }

    v_MobilityControlInfo := f_Generate_cs_MobilityControlInfo_HO_CRNTI(p_TargetCellId, p_C_RNTI, v_CarrierFreqEutra, v_CarrierBandwidthEUTRA, v_PucchConfigCommon_Needed);   // @sic R5s120209 change 1 sic@

    v_RadioResourceConfigDedicated := f_Generate_cs_508_RadioResourceConfigDedicated_HO (p_TargetCellId, p_Dsr_TransMax);

    // Build DRB identity list from inputed radiobearer config list

    v_DRB_IdentityList := fl_EUTRA_DRB_IdentityList(p_DrbConfigList);

    // Step 1: Transfer of the PDCP Count for AM DRBs from source to target cell   @sic R5-120721 change 7 sic@

    f_EUTRA_SS_PdcpCount_Handover(p_SourceCellId, p_TargetCellId, v_DRB_IdentityList.AMonly);

    // Step 2: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverInit_REQ(p_SourceCellId, p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 3: Configure C-RNTI based contention resolution in target cell

    f_EUTRA_SS_ConfigRachProcedure_HO(p_TargetCellId, cs_TimingInfo_Now, p_C_RNTI);

    // Save new C-RNTI in cell configuration for p_TargetCellId

    f_EUTRA_CellInfo_SetC_RNTI ( p_TargetCellId, p_C_RNTI );

    // Step 4: activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( p_TargetCellId,

                                                               v_Auth_Params,

                                                               v_NextHopChainingCount,

                                                               -,

                                                               v_DRB_IdentityList.AMonly);

    f_EUTRA_Security_Set ( v_Auth_Params );

    // Step 5: configure DRX and MeasGap at the target cell (if necessary)

    if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourceCellId)) {

      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_SourceCellId);

      f_EUTRA_SS_DrxCtrlConfig(p_TargetCellId, v_DrxCtrl);

    }

    if (v_MeasGapConfigIsPresent) {

      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetCellId, cs_MeasGapCtrl_Config(p_MeasConfig.measGapConfig));

    }

    // Step 6: stop periodic TA on source cell   @sic R5s100520 change 10 sic@

    f_EUTRA_SS_CommonCellConfig(p_SourceCellId, cas_ULGrantAllocation_REQ(p_SourceCellId, cs_TimingInfo_Now, -, cs_PUCCH_Synch_None, p_UL_GrantConfig_SourceCell));

    // Step 7: configure UL grant configuration on target cell    @sic R5s100515 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_ULGrantAllocation_REQ(p_TargetCellId, cs_TimingInfo_Now, p_DciUlInfo_TargetCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_TargetCell));

    // Step 8: Transmit RRCConnectionReconfiguration/HO command to UE

    f_EUTRA_RRC_ConnReconfigHO_Common(p_SourceCellId,

                                      p_MeasConfig,

                                      v_MobilityControlInfo,

                                      v_RadioResourceConfigDedicated,

                                      cs_508SecurityConfigHO_IntraLTE);

    // Step 9:

    SRB.receive(car_SRB1_RrcPdu_IND(p_TargetCellId, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));       // @sic R5s100430  Change 2 sic@

    // Step 10: configure periodic TA on target cell    @sic R5s100525 change 1.3 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_PUCCH_Synch_Config_REQ(p_TargetCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def, tsc_NoCnfReq)); // @sic R5s100801 sic@

    // Step 11: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverComplete_REQ(p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 12: configure RACH procedure for target cell back to default    @sic R5s100520 change 11 sic@

    f_EUTRA_SS_ConfigRachProcedure_Def(p_TargetCellId, -, -, -, -, -, v_CnfFlag);     /* @sic acc. to R5s100029 change 3 sic@ */

    // Step 13: Re-establish SRBs and DRB(s) in source cell

    f_EUTRA_SS_SRBs_DRBs_Release ( p_SourceCellId, -, v_DRB_IdentityList.All, v_CnfFlag); // @sic R5s100801 sic@

    f_EUTRA_SS_SRBs_DRBs_Config ( p_SourceCellId, -, p_DrbConfigList, v_CnfFlag); // @sic R5s100801 sic@

    // Step 14: Release DRX and MeasGap configuration

    f_EUTRA_SS_DrxCtrlConfig(p_SourceCellId, cs_DrxCtrl_None, -, v_CnfFlag);  // @sic R5s100801 sic@

    f_EUTRA_SS_MeasGapCtrlConfig(p_SourceCellId, cs_MeasGapCtrl_None, -, v_CnfFlag);  // @sic R5s110743 sic@

  }


After change:

	function f_EUTRA_508RRC_IntraLTE_HO_InterCell ( CellId_Type                           p_SourceCellId,

                                                  CellId_Type                           p_TargetCellId,

                                                  template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def,

                                                  C_RNTI                                p_C_RNTI        := tsc_C_RNTI_Def2,

                                                  template (omit)  MeasConfig           p_MeasConfig    := omit,

                                                  template (omit)  DciUlInfo_Type       p_DciUlInfo_TargetCell      := cs_DciInfo_CcchDcchDtchUL_Def,

                                                  template (value) UL_GrantConfig_Type  p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_Def,

                                                  template (value) UL_GrantConfig_Type  p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_Def,

                                                  Dsr_TransMax_Type                     p_Dsr_TransMax := n4,

                                                  template (omit) CarrierFreqEUTRA      p_CarrierFreqEutra := omit,

                                                  template (omit) boolean p_CnfFlag := false )  runs on EUTRA_PTC      //Anritsu PTO TTCN CR required

  { /* @sic R5s100491: enhancement to support MeasConfig sic@ */

    /* @sic R5s100524: common change of HO functions sic@ */

    /* @sic R5-106063: Sequence of inter-cell handover sic@ */

    /* @sic R5s100801: No CNF for SS configuration of steps 10..14 sic@

     *                 when there is data pending at the UE the UE may send this data immediately after the HO;

     *                 this would collide with CNFs sent during reconfigurations of source and target call after the HO;

     *                 on the other hand none of the reconfigurations in source or target cell should have impact in cases the test case sends data after the HO

     *                 => reconfiguration steps 10..14 are done without request of CNF as proposed in the CR */

    /* @sic R5s100772: p_UL_GrantConfig_SourceCell sic@ */

    /* @sic R5s110168: additional parameter p_Dsr_TransMax sic@ */

    /* @sic R5-112122: UL_DL_Earfcn is omit in case of inter-cell HO when there is no change of frequency sic@ */

    /* @sic R5s110279 change 1: new optional parameter p_CarrierFreqEutra to force the CarrierFreqEutra e.g. to omit the ul-CarrierFreq as requested for TC_8_2_4_9 sic@ */

    /* @sic R5s110743, R5s120039: handling of MeasGapConfig sic@ */

    /* @sic R5s120057, R5-120677: In case of multiband scenarios the bandwidth can be different for source and target cell sic@ */

    /* @sic R5-120721 change 7: handover of PDCP Count for AM DRBs from source cell to target cell sic@ */

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment

    var Frequency_IE_Type v_Frequency_IE_SourceCell := f_EUTRA_CellInfo_GetFrequencyIEs (p_SourceCellId);

    var Frequency_IE_Type v_Frequency_IE_TargetCell := f_EUTRA_CellInfo_GetFrequencyIEs (p_TargetCellId);

    var CarrierFreqEUTRA v_CarrierFreqEutra_SourceCell := v_Frequency_IE_SourceCell.UL_DL_Earfcn;

    var CarrierFreqEUTRA v_CarrierFreqEutra_TargetCell := v_Frequency_IE_TargetCell.UL_DL_Earfcn;

    var template (omit) CarrierFreqEUTRA v_CarrierFreqEutra := omit;

    var template (omit) CarrierBandwidthEUTRA v_CarrierBandwidthEUTRA := omit;

    var template (value) MobilityControlInfo v_MobilityControlInfo;

    var template (value) RadioResourceConfigDedicated v_RadioResourceConfigDedicated := f_Generate_cs_508_RadioResourceConfigDedicated_HO (p_TargetCellId, p_Dsr_TransMax);

    var boolean v_MeasGapConfigIsPresent := isvalue(p_MeasConfig) and ispresent(p_MeasConfig.measGapConfig);

    var boolean v_PucchConfigCommon_Needed := false;

    var DrxCtrl_Type v_DrxCtrl;

    if (isvalue(p_CarrierFreqEutra)) {      // use the give CarrierFreqEutra; @sic R5s110279 change 1 sic@

      v_CarrierFreqEutra := p_CarrierFreqEutra;

    }

    else {                                  // => determine the CarrierFreqEutra depending on cell information stored for sourece and target cell

      if (v_CarrierFreqEutra_SourceCell != v_CarrierFreqEutra_TargetCell) {    // only if CarrierFreqEUTRA is different in both cells the CarrierFreqEUTRA of the target cell is handed over @sic R5-112122 sic@

        v_CarrierFreqEutra := v_CarrierFreqEutra_TargetCell;

      }

    }

    if (v_Frequency_IE_SourceCell.DL_ChBandwidth != v_Frequency_IE_TargetCell.DL_ChBandwidth) {                         // @sic R5s120057 sic@

      v_PucchConfigCommon_Needed := true;                                                                               // @sic R5s120209 change 1 sic@

      v_CarrierBandwidthEUTRA := cs_CarrierBandwidthEUTRA(v_Frequency_IE_TargetCell.DL_ChBandwidth);

    }

    v_MobilityControlInfo := f_Generate_cs_MobilityControlInfo_HO_CRNTI(p_TargetCellId, p_C_RNTI, v_CarrierFreqEutra, v_CarrierBandwidthEUTRA, v_PucchConfigCommon_Needed);   // @sic R5s120209 change 1 sic@

    v_RadioResourceConfigDedicated := f_Generate_cs_508_RadioResourceConfigDedicated_HO (p_TargetCellId, p_Dsr_TransMax);

    // Build DRB identity list from inputed radiobearer config list

    v_DRB_IdentityList := fl_EUTRA_DRB_IdentityList(p_DrbConfigList);

    // Step 1: Transfer of the PDCP Count for AM DRBs from source to target cell   @sic R5-120721 change 7 sic@

    f_EUTRA_SS_PdcpCount_Handover(p_SourceCellId, p_TargetCellId, v_DRB_IdentityList.AMonly);

    // Step 2: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverInit_REQ(p_SourceCellId, p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 3: Configure C-RNTI based contention resolution in target cell

    f_EUTRA_SS_ConfigRachProcedure_HO(p_TargetCellId, cs_TimingInfo_Now, p_C_RNTI);

    // Save new C-RNTI in cell configuration for p_TargetCellId

    f_EUTRA_CellInfo_SetC_RNTI ( p_TargetCellId, p_C_RNTI );

    // Step 4: activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( p_TargetCellId,

                                                               v_Auth_Params,

                                                               v_NextHopChainingCount,

                                                               -,

                                                               v_DRB_IdentityList.AMonly);

    f_EUTRA_Security_Set ( v_Auth_Params );

    // Step 5: configure DRX and MeasGap at the target cell (if necessary)

    if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourceCellId)) {

      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_SourceCellId);

      f_EUTRA_SS_DrxCtrlConfig(p_TargetCellId, v_DrxCtrl);

    }

    if (v_MeasGapConfigIsPresent) {

      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetCellId, cs_MeasGapCtrl_Config(p_MeasConfig.measGapConfig));

    }

    // Step 6: stop periodic TA on source cell   @sic R5s100520 change 10 sic@

    f_EUTRA_SS_CommonCellConfig(p_SourceCellId, cas_ULGrantAllocation_REQ(p_SourceCellId, cs_TimingInfo_Now, -, cs_PUCCH_Synch_None, p_UL_GrantConfig_SourceCell));

    // Step 7: configure UL grant configuration on target cell    @sic R5s100515 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_ULGrantAllocation_REQ(p_TargetCellId, cs_TimingInfo_Now, p_DciUlInfo_TargetCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_TargetCell));

    // Step 8: Transmit RRCConnectionReconfiguration/HO command to UE

    f_EUTRA_RRC_ConnReconfigHO_Common(p_SourceCellId,

                                      p_MeasConfig,

                                      v_MobilityControlInfo,

                                      v_RadioResourceConfigDedicated,

                                      cs_508SecurityConfigHO_IntraLTE);

    // Step 9:

    SRB.receive(car_SRB1_RrcPdu_IND(p_TargetCellId, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));       // @sic R5s100430  Change 2 sic@

    // Step 10: configure periodic TA on target cell    @sic R5s100525 change 1.3 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_PUCCH_Synch_Config_REQ(p_TargetCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def, tsc_NoCnfReq)); // @sic R5s100801 sic@

    // Step 11: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverComplete_REQ(p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 12: configure RACH procedure for target cell back to default    @sic R5s100520 change 11 sic@

    f_EUTRA_SS_ConfigRachProcedure_Def(p_TargetCellId, -, -, -, -, -, p_CnfFlag);     /* @sic acc. to R5s100029 change 3 sic@ */

    // Step 13: Re-establish SRBs and DRB(s) in source cell

    f_EUTRA_SS_SRBs_DRBs_Release ( p_SourceCellId, -, v_DRB_IdentityList.All, p_CnfFlag); // @sic R5s100801 sic@

    f_EUTRA_SS_SRBs_DRBs_Config ( p_SourceCellId, -, p_DrbConfigList, p_CnfFlag); // @sic R5s100801 sic@

    // Step 14: Release DRX and MeasGap configuration

    f_EUTRA_SS_DrxCtrlConfig(p_SourceCellId, cs_DrxCtrl_None, -, p_CnfFlag);  // @sic R5s100801 sic@

    f_EUTRA_SS_MeasGapCtrlConfig(p_SourceCellId, cs_MeasGapCtrl_None, -, p_CnfFlag);  // @sic R5s110743 sic@

  }

